Задача 29

( Задача 9.65. Сборник задач по теоретической механике/ Под ред. К.С. Колесникова. М.: Наука, Главная редакция физико-математической литературы, 1983. – 320 с. )
Треугольник ОАВ, стороны которого представляют собой однородные тонкие стержни равной длины, массы m каждый, может вращаться вокруг горизонтальной оси, проходящей через вершину О перпендикулярно плоскости ОАВ. Из положения, когда сторона ОА горизонтальна, треугольник отпускают без начальной скорости.

Определить давление на ось О в начальный момент движения и в момент, когда сторона АВ станет горизонтальной.

Для решения задачи использовать следующие значения параметров:

m = 1 кг.

Точный теоретический ответ

Для принятых значений параметров натяжения равны

R1 =16.99141842 Н;
R2 = 49.05000000 Н.

Решение задачи в EULER

Система состоит из двух звеньев.

· Инерциальное звено (Неподвижная_точка). В проекте звено отображается точкой O.

· Рамка (Треугольник). Звено отображается цилиндрами solid1, solid2, solid3, которые имеют массу m.

Звенья Неподвижная точка и Треугольник соединены шарниром типа пара вращения.. Для определения сил создан датчики силы F. Для определения R1 нужно не запуская динамики нажать расчет сил и ускорений. Для определения R2 создано событие event1. В проекте задана гравитация с ускорением свободного падения 9.81 [m/ s2].

Результаты моделирования

	Постоянный шаг интегрирования
	Точный теоретический ответ

R2[N]
	Ответ, полученный при моделировании

R2[N]
	Относительное отличие от теоретического решения
T2[-]

	0,05
	49.05000000
	49.86639831
	0.01664421

	0,01
	49.05000000
	49.05004189
	0.00000085


	Точный теоретический ответ

R1[N]
	Ответ, полученный при моделировании

R1[N]
	Относительное отличие от теоретического решения

T1[-]

	16.99141842
	16.99141846
	2.499e-009


Текст проекта в EULER
point O = point( 0 [ m ], 0 [ m ], 0 [ m ] );

point A = point( -1  [ m ], 0 [ m ], 0 [ m ] );

point B = point( -0.5 [ m ], -sin(60[deg])*1[ m ], 0 [ m ] );

scalar r = 0.0001[ m ];

scalar m = 1[ kg ];

solid solid1 = cylinder( O, A, r, mass = m );

solid solid2 = cylinder( A, B, r, mass = m );

solid solid3 = cylinder( B, O, r, mass = m );

body Неподвижная_точка = body( color = RGB( 229, 229, 229 ) );

color Green = RGB( 30, 206, 38 );

body Треугольник = body( color = Green );

gravity gravity1 = parallel( reverse( projectY ) );

joint joint1 = rotational( Неподвижная_точка, Треугольник, O, projectZ );

sensor Fx = jointForce( force:, Неподвижная_точка, O, projectX, joint1 );

sensor Fy = jointForce( force:, Неподвижная_точка, O, projectY, joint1 );

sensor F = sqrt(Fx*Fx+Fy*Fy);

vector AO = vectorPP( A, O );

event event1 = reformsBySensor( list(  ), joint1.gamma, 60[ deg ], gauge = list( F ) );

sensor R1 = m*9.81*sqrt(3)*1[N/kg];

sensor R2 = m*9.81*5[N/kg];

sensor T1 = abs((F-R1)/R1);

sensor T2 = abs((event1.F-R2)/R2);

/\///////////////////////////////////////////////////////////////////////////////////

/\   Прикрепление объектов

body Неподвижная_точка < (O );

body Треугольник < (solid1, solid2, solid3, A, B );

/\///////////////////////////////////////////////////////////////////////////////////

/\   Инерциальное звено;

set ground = Неподвижная_точка;

/\///////////////////////////////////////////////////////////////////////////////////

/\   Единицы измерения;

set units = SI;

